The New Bisection Plus Algorithm
by
Namir Shammas

Introduction
This article presents a new variant for the root-bracketing Bisection algorithm. This new version injects a lot of punch to the Bisection method which is the slowest root-seeking method.

The Bisection Algorithm
There are numerous algorithms that calculate the roots of single-variable nonlinear functions. The most popular of such algorithms is Newton’s method. The slowest and simplest root seeking algorithm is the Bisection method. This method has the user select an interval that contains the sought root. The method iteratively shrinks the root-bracketing interval to zoom in on the sought root. Here is the pseudo-code for the Bisection algorithm:

Given \( f(x) = 0 \), the root-bracketing interval \([A, B]\), and the tolerance for the root of \( f(x) \):

- Calculate \( Fa = f(A) \) and \( Fb = f(B) \).
- Exit if \( Fa \cdot Fb > 0 \).
- Repeat
  - \( X = (A + B) / 2 \)
  - \( Fx = f(X) \)
  - If \( Fx \cdot Fa > 0 \) then
    - \( A = X \)
    - \( Fa = Fx \)
  - Else
    - \( B = X \)
    - \( Fb = Fx \)
  - End
- Until \( |A - B| < \) tolerance
- Return root as \((A + B) / 2\)

The above pseudo-code shows how the algorithm iteratively halves the root-bracketing until it zooms on the root. The Bisection method is the slowest...
converging method. It’s main virtue is that it is guaranteed to work if \( f(x) \) is continuous in the interval \([A, B]\) and \( f(A) \times f(B) \) is negative.

**Newton’s Method**

I will also compare the new algorithm with Newton’s method. This comparison serves as an upper limit test. I am implementing Newton’s method based on the following pseudo-code:

Given \( f(x) = 0 \), the root-bracketing interval \([A, B]\), and the tolerance for the root of \( f(x) \):

- Calculate \( X = (A+B)/2 \)
- Repeat
  - \( h = 0.001 \times (\sqrt{|X|} + 1) \)
  - \( Fx = f(X) \)
  - \( \text{Diff} = h \times Fx / (f(X+h) - Fx) \)
  - \( X = X - \text{Diff} \)
- Until \( |\text{Diff}| < \text{tolerance} \)
- Return root as \( X \)

The above code shows that the implementation of Newton’s method starts with the same interval \([A, B]\) that is already available for the Bisection and Bisection Plus methods. Thus, the algorithm derives its single initial guess as the midpoint of that interval.

**The Bisection Plus Algorithm**

In my effort to enhance the Bisection method, I started with several approaches that takes the midpoint in the original Bisection method and enhances it. I initially worked with adding some random fluctuation to the midpoint value in \([A, B]\). This effort showed little promise.

I went back to the proverbial drawing board. The essential part of Bisection method is that it limits itself to comparing the signs of \( f(x) \) values. A few years ago, I developed the Quartile Method\(^3\), which improves on the Bisection method by comparing the absolute values of \( f(A) \) and \( f(B) \) in order to get a better “midpoint” value. In designing the Bisection Plus algorithm, I decided to up the ante and work with the values of \( f(A) \), \( f(B) \), and \( f((A+B)/2)) \). The new algorithm has the following steps:

1. Each iteration of the Bisection Plus method calculates the midpoint, call it \( X_1 \), and its associated function value \( f(X_1) \).
2. The new algorithm then calculates the slope and intercept passing through $X_1$ and either A or B. The method selects the interval endpoint whose function value has the opposite sign of $f(X_1)$.

3. Using this new line, the algorithm calculates $X_2$—a better estimate for the root. The choice of using the proper endpoint and $X_1$ ensures that $X_2$ lies in the interval $[A, B]$. The algorithm also calculates $f(X_2)$.

4. Each iteration ends up with the original interval-defining values A and B, and two new values, $X_1$ and $X_2$, within that interval. The process of shrinking the root-bracketing interval involves two tests:
   a. If $f(X_1) \times f(X_2)$ is negative, then the interval $[X_1, X_2]$ replaces the interval $[A, B]$ causing a quick reduction in the root-bracketing interval.
   b. If $f(X_1) \times f(X_2)$ is positive, then the method uses the value of $f(A) \times f(X_2)$ to determine which of A or B is replaced by $X_2$.

An additional improvement to the algorithm, inspired by the False-Position method, compares the newly calculated $X_2$ value with the one from the previous iteration. If the two values are within the tolerance limit, the algorithm stops iterating.

It is worth pointing out that while both Newton’s method and the Bisection Plus calculate slope values, these values are different in context. Newton’s method calculate the tangent of $f(X)$ at $X$. By contrast, the Bisection Plus method calculate the slope of the straight line between $X_1$ and one of the endpoints. Thus, the Bisection Plus method is not vulnerable to values of $X$ where the derivative of $f(X)$ is near zero, as is the case with Newton’s method.

Let me present the pseudo-code for the Bisection Plus method:

**Given** $f(x)=0$, the root-bracketing interval $[A,B]$, and the tolerance for the root of $f(x)$:

- Calculate $F_a = f(A)$ and $F_b = f(B)$.
- Exit if $F_a \times F_b > 0$
- Last$X = A$
- Repeat
  - $X_1 = (A+B)/2$
  - $F(x_1) = f(X_1)$
  - If $F(x_1) \times F_a > 0$ then
    - Slope = $(F_b - F(x_1))/(B - X_1)$
    - Intercept = $F_b - \text{Slope} \times B$
  - Else
The New Bisection Plus Algorithm

- \[ \text{Slope} = (F_a - F_{x1})/(A - 1) \]
- \[ \text{Intercept} = F_a - \text{Slope} \times A \]
  
  \[ \text{End} \]
  
  \[ \text{X2} = -\text{Intercept} / \text{Slope} \]
  
  \[ F_{x2} = f(X2) \]
  
  \[ \text{If} \ F_{x1} \times F_{x2} < 0 \text{ then} \]
  
  \[ \text{A} = X_1 \]
  
  \[ F_a = F_{x1} \]
  
  \[ B = X_2 \]
  
  \[ F_b = F_{x2} \]
  
  \[ \text{Else} \]
  
  \[ \text{If} \ F_{x2} \times F_a > 0 \text{ then} \]
  
  \[ A = X_2 \]
  
  \[ F_a = F_{x2} \]
  
  \[ \text{Else} \]
  
  \[ B = X_2 \]
  
  \[ F_b = F_{x2} \]
  
  \[ \text{End} \]
  
  \[ \text{End} \]
  
  \[ \text{If} \ |X_2 - \text{Last}X| < \text{tolerance} \text{ then exit loop} \]
  
  \[ \text{Last}X = X_2 \]
  
  - Until \ |A - B| < \text{tolerance} \]
  
  - Return \ X_2 \]

Figure 1 depicts an iteration of the Bisection Plus algorithm.
Figure 1. The steps within an iteration of the Bisection Plus Algorithm.

Figure 1 depicts an iteration that ends up replacing the initial root-bracketing interval \([A, B]\) with \([X_2, B]\). If \(f(X_2)\) was positive, the iteration would have replaced the initial root-bracketing interval \([A, B]\) with the narrower interval \([X_1, X_2]\). From the testing that I have done, the latter occurs frequently and helps to quickly narrow the root-bracketing interval around the targeted root value.

Figure 2 shows a case where some of the values of \(|f(X)|\) for \(X\) in \([A, B]\) are greater than \(|f(A)|\) and \(|f(B)|\). The figure illustrates why \(X_2\) is calculated using \(X_1\) and either endpoint whose function value has the opposite sign of \(f(X_1)\)—in the case of Figure 2, using \(X_1\) and \(A\). If \(X_2\) is calculated using \(X_1\) and \(B\), then the value of \(X_2\) lands outside the root-bracketing interval \([A, B]\). Using the scheme that I suggested to calculate \(X_2\) simplify matters, because the algorithm needs not check if any \(X\) in \([A, B]\) has function values that exceed \(|f(A)|\) and/or \(|f(B)|\).

Testing with Excel VBA Code

I tested the new algorithm using Excel taking advantage of the application’s worksheet for easy input and the display of intermediate calculations. The following listing shows the Excel VBA code used for testing:

Option Explicit
Function MyFx(ByVal sFx As String, ByVal X As Double) As Double
    sFx = UCase(Trim(sFx))
    sFx = Replace(sFx, "$X", "(" & X & ")")
    MyFx = Evaluate(sFx)
End Function

Sub Go()
    ' Bisection Plus algorithm
    ' Version 1.00  1/9/2014
    ' Copyright (c) 2014 Namir Clement Shammas
    Dim R As Integer, Count As Long, NumFxCalls As Integer
    Dim A As Double, B As Double, X As Double, LastX As Double
    Dim X1 As Double, X2 As Double, FX1 As Double, FX2 As Double
    Dim FA As Double, FB As Double, FX As Double, Toler As Double
    Dim Slope As Double, Intercept As Double
    Dim h As Double, Diff As Double
    Dim sFx As String
    Range("B3:Z1000").Value = ""
    A = [A2].Value
    B = [A4].Value
    Toler = [A6].Value
    sFx = [A8].Value
    FA = MyFx(sFx, A)
    FB = MyFx(sFx, B)
    If FA * FB > 0 Then
        MsgBox "F(A) & F(B) have the same signs"
        Exit Sub
    End If
    ' Bisection
    R = 3
    NumFxCalls = 2
    Do While Abs(A - B) > Toler
        NumFxCalls = NumFxCalls + 1
        X = (A + B) / 2
        FX = MyFx(sFx, X)
        If FX * FA > 0 Then
            A = X
            FA = FX
        Else
            B = X
            FB = FX
        End If
        Cells(R, 2) = A
Cells(R, 3) = B
R = R + 1
Loop
Cells(R, 2) = (A + B) / 2
Cells(R, 3) = "FX Calls=" & NumFxCalls

' Bisection Plus
A = [A2].Value
B = [A4].Value
FA = MyFx(sFx, A)
FB = MyFx(sFx, B)
R = 3
LastX = A
NumFxCalls = 2
Do
  X1 = (A + B) / 2
  FX1 = MyFx(sFx, X1)
  NumFxCalls = NumFxCalls + 1
  If FA * FX1 > 0 Then
    Slope = (FB - FX1) / (B - X1)
    Intercept = FB - Slope * B
  Else
    Slope = (FA - FX1) / (A - X1)
    Intercept = FA - Slope * A
  End If
  X2 = -Intercept / Slope
  FX2 = MyFx(sFx, X2)
  NumFxCalls = NumFxCalls + 1
  If FX1 * FX2 < 0 Then
    A = X1
    FA = FX1
    B = X2
    FB = FX2
    Cells(R, 6) = "New interval"
  Else
    If FA * FX2 > 0 Then
      A = X2
      FA = FX2
    Else
      B = X2
      FB = FX2
    End If
  End If
End If
Cells(R, 4) = A
Cells(R, 5) = B
If Abs(LastX - X2) < Toler Then Exit Do
LastX = X2
R = R + 1
Loop Until Abs(A - B) < Toler
Cells(R, 4) = X2
Cells(R, 5) = "FX Calls=" & NumFxCalls

' Newton's method
A = [A2].Value
B = [A4].Value
X = (A + B) / 2
R = 3
NumFxCalls = 2
Diff = 10 * Toler
Do
  h = 0.001 * (Abs(X) + 1)
  FX = MyFx(sFx, X)
  Diff = h * FX / (MyFx(sFx, X + h) - FX)
  X = X - Diff
  NumFxCalls = NumFxCalls + 2
  Cells(R, 7) = X
  Cells(R, 8) = Diff
  R = R + 1
Loop Until Abs(Diff) <= Toler
Cells(R, 7) = X
Cells(R, 8) = "FX Calls=" & NumFxCalls
End Sub

The VBA function MyFX calculates the function value based on a string that contains the function’s expression. This expression must use $X as the variable name. Using function MyFX allows you to specify the function f(X)=0 in the spreadsheet and not hard code it in the VBA program. Granted that this approach trades speed of execution with flexibility. However, with most of today’s PCs you will hardly notice the difference in execution speeds.

The subroutine Go performs the root-seeking calculations that compare the Bisection method, Bisection Plus method, and Newton’s method. Figure 2 shows a snapshot of the Excel spreadsheet used in the calculations for the Bisection, Bisection Plus, and Newton’s method.
The New Bisection Plus Algorithm

Figure 2. The Excel spreadsheet used to compare the Bisection, Bisection Plus, and Newton’s method.

The Input Cells

The VBA code relies on the following cells to obtain data:

- Cells A2 and A4 supply the values for the root-bracketing interval [A, B].
- Cell A6 contains the tolerance value.
- Cell A8 contains the expression for f(X)=0. Notice that the contents of cell A8 use $X$ as the variable name. The expression is case insensitive.

Output

The spreadsheet displays output in the following three sets of columns:

- Columns B and C display the updated values for the root-bracketing interval [A, B] for the Bisection method. This interval shrinks with each iteration until the Bisection method zooms on the root. The bottom most value, in column B, is the best estimate for the root. To its right is the total number of function calls made during the iterations.

- Columns D, and E display the updated values for the root-bracketing interval [A, B] for the Bisection Plus method. Column F displays comments made by the code that point out when $X_1$ and $X_2$ form a new root-bracketing interval. The bottom most value, in column D, is the best estimate for the root. To its right is the total number of function calls made during the iterations.
• Columns G and H display the refined guess for the root and the refinement value, respectively, using Newton’s method. The bottom most value, in column G, is the best estimate for the root. To its right is the total number of function calls made during the iterations.

The Results
My aim is to significantly accelerate the Bisection Plus method compared to the Bisection method. I do not expect the Bisection Plus method to outperform Newton’s method. The results proved me partial wrong regarding the last statement. Table 1 shows a summary of the results. The metrics for comparing the algorithms include the number of iterations and, perhaps more importantly, the number of function calls. I consider the number of function calls as the underlying cost of doing business, so to speak. I have come across new root-seeking algorithms that require fewer iterations that popular algorithms like Newton’s method and Halley’s method. However, these new algorithms require more function calls to zoom in on the root in fewer iterations.

<table>
<thead>
<tr>
<th>Function</th>
<th>[A, B]</th>
<th>Toler</th>
<th>Root</th>
<th>Iterations</th>
<th>Num Fx Calls</th>
</tr>
</thead>
<tbody>
<tr>
<td>Exp(x)-4*x^2</td>
<td>[3, 5]</td>
<td>1E-8</td>
<td>4.30658</td>
<td>Bisec= 28</td>
<td>Bisec= 30</td>
</tr>
<tr>
<td></td>
<td></td>
<td></td>
<td></td>
<td>Bisec+ = 7</td>
<td>Bisec+ = 18</td>
</tr>
<tr>
<td></td>
<td></td>
<td></td>
<td></td>
<td>Newton=6</td>
<td>Newton=14</td>
</tr>
<tr>
<td>Exp(x)-3*x^2</td>
<td>[1, 4]</td>
<td>1E-8</td>
<td>3.73307</td>
<td>Bisec= 29</td>
<td>Bisec= 31</td>
</tr>
<tr>
<td></td>
<td></td>
<td></td>
<td></td>
<td>Bisec+ = 7</td>
<td>Bisec+ = 18</td>
</tr>
<tr>
<td></td>
<td></td>
<td></td>
<td></td>
<td>Newton=10</td>
<td>Newton=22</td>
</tr>
<tr>
<td>Exp(x)-3*x^2</td>
<td>[3, 4]</td>
<td>1E-8</td>
<td>3.73307</td>
<td>Bisec= 27</td>
<td>Bisec= 29</td>
</tr>
<tr>
<td></td>
<td></td>
<td></td>
<td></td>
<td>Bisec+ = 6</td>
<td>Bisec+ = 16</td>
</tr>
<tr>
<td></td>
<td></td>
<td></td>
<td></td>
<td>Newton=6</td>
<td>Newton=14</td>
</tr>
<tr>
<td>(X-2.345) * (X-12.345) *</td>
<td>[1, 11]</td>
<td>1E-8</td>
<td>2.345</td>
<td>Bisec= 30</td>
<td>Bisec= 32</td>
</tr>
<tr>
<td>(X-23.456)</td>
<td></td>
<td></td>
<td></td>
<td>Bisec+ = 5</td>
<td>Bisec+ = 14</td>
</tr>
<tr>
<td></td>
<td></td>
<td></td>
<td></td>
<td>Newton=9</td>
<td>Newton=20</td>
</tr>
<tr>
<td>(X-23.456)</td>
<td></td>
<td></td>
<td></td>
<td>Bisec+ = 4</td>
<td>Bisec+ = 12</td>
</tr>
<tr>
<td></td>
<td></td>
<td></td>
<td></td>
<td>Newton=6</td>
<td>Newton=14</td>
</tr>
<tr>
<td>LN(X^4)-X</td>
<td>[8,9]</td>
<td>1E-8</td>
<td>8.66125</td>
<td>Bisec= 27</td>
<td>Bisec= 29</td>
</tr>
<tr>
<td></td>
<td></td>
<td></td>
<td></td>
<td>Bisec+ = 4</td>
<td>Bisec+ = 12</td>
</tr>
<tr>
<td></td>
<td></td>
<td></td>
<td></td>
<td>Newton=4</td>
<td>Newton=10</td>
</tr>
<tr>
<td>Cos(X)-X</td>
<td>[0.5,1]</td>
<td>1E-8</td>
<td>0.73908</td>
<td>Bisec= 26</td>
<td>Bisec= 28</td>
</tr>
<tr>
<td></td>
<td></td>
<td></td>
<td></td>
<td>Bisec+ = 3</td>
<td>Bisec+ = 10</td>
</tr>
<tr>
<td></td>
<td></td>
<td></td>
<td></td>
<td>Newton=4</td>
<td>Newton=10</td>
</tr>
</tbody>
</table>
Table 1. Summary of the results comparing the Bisection, Bisection Plus, and Newton's method.

The above table shows that the Bisection Plus method consistently outperforms the Bisection method. The table also highlights in red the cases where the Bisection Plus method performs better than Newton’s method. Of course there is a huge number of test cases that vary the tested function and root-bracketing range. Due to time limitation, I have chosen the above few test cases which succeeded in proving my goals.

Conclusion
The Bisection Plus algorithm offers significant improvement over the Bisection method. The new algorithm has an efficiency that is somewhat comparable to that of Newton’s method.

References

Document Information

<table>
<thead>
<tr>
<th>Version</th>
<th>Date</th>
<th>Comments</th>
</tr>
</thead>
<tbody>
<tr>
<td>1.0.0</td>
<td>1/9/2014</td>
<td>Initial release.</td>
</tr>
<tr>
<td>1.0.1</td>
<td>1/10/2014</td>
<td>Edit of general description of algorithm</td>
</tr>
<tr>
<td>1.0.2</td>
<td>1/18/2014</td>
<td>Minor edits.</td>
</tr>
</tbody>
</table>